Hecke Grossencharacters

A GP tutorial

A. Page

IMB
Inria/Université de Bordeaux

11/01/2024

Black-box definition

K number field of degree n and signature (r_1, r_2).

The "group of idèles of K" is a topological Abelian group $\mathbb{A}_K^\times$ with

- an embedding $K_v^\times \hookrightarrow \mathbb{A}_K^\times$ for every completion K_v of K;
- a diagonal embedding $K^\times \hookrightarrow \mathbb{A}_K^\times$.

The quotient ("idèle class group")

$$C_K = \mathbb{A}_K^\times / K^\times$$

is isomorphic to $\mathbb{R} \times$ a compact group.

A **Hecke character** is a continuous morphism

$$\chi \colon C_K \to \mathbb{C}^\times.$$

Finite level version

The groups C_K or $\mathrm{Hom}(C_K, \mathbb{C}^\times)$ are too big to handle algorithmically: cut them into smaller pieces!

Modulus $\mathfrak{m}$: pair $(\mathfrak{m}_f, \mathfrak{m}_\infty) =$ (nonzero ideal, subset of the real embeddings).

We can define certain open subgroups $U(\mathfrak{m})$ of $\mathbb{A}_K^\times$ such that

- every Hecke character vanishes on some $U(\mathfrak{m})$, and
- $C_\mathfrak{m} = \mathbb{A}_K^\times / K^\times U(\mathfrak{m})$ is of an appropriate size: a finite dimensional manifold.

$$1 \to \mathbb{R} \times \text{compact torus} \to C_\mathfrak{m} \to \text{finite group} \to 1.$$

Finite level version

The groups C_K or $\mathrm{Hom}(C_K, \mathbb{C}^\times)$ are too big to handle algorithmically: cut them into smaller pieces!

Modulus $\mathfrak{m}$: pair $(\mathfrak{m}_f, \mathfrak{m}_\infty) =$ (nonzero ideal, subset of the real embeddings).

We can define certain open subgroups $U(\mathfrak{m})$ of $\mathbb{A}_K^\times$ such that

- every Hecke character vanishes on some $U(\mathfrak{m})$, and
- $C_\mathfrak{m} = \mathbb{A}_K^\times / K^\times U(\mathfrak{m})$ is of an appropriate size: a finite dimensional manifold.

$$1 \to \left[(\mathbb{R}_{>0})^{r_1} \times (\mathbb{C}^\times)^{r_2}\right] / \left[\mathbb{Z}_K^\times \cap U(\mathfrak{m})\right] \to C_\mathfrak{m} \to \mathrm{Cl}_\mathfrak{m}(K) \to 1.$$

Finite level version

For Hecke characters, this means:

$$\mathrm{Hom}(C_F, \mathbb{C}^\times) = \bigcup_{\mathfrak{m}} \mathrm{Hom}(C_{\mathfrak{m}}, \mathbb{C}^\times),$$

and for every $\mathfrak{m}$,

$$\mathrm{Hom}(C_{\mathfrak{m}}, \mathbb{C}^\times) \cong \text{finite} \times \mathbb{Z}^{n-1} \times \mathbb{C}.$$

Finite order characters of $C_{\mathfrak{m}}$ are exactly characters of $\mathrm{Cl}_{\mathfrak{m}}(K)$.

Initialisation

We initialise $\mathrm{Hom}(C_{\mathfrak{m}}, \mathbb{C}^\times)$ with `gcharinit`:

```
? bnf = bnfinit(polcyclo(5),1);
? gc = gcharinit(bnf,5);
? gc.cyc
% = [5, 0, 0, 0, 0.E-57]
```

$$\mathrm{Hom}(C_{\mathfrak{m}}, \mathbb{C}^\times) \cong \mathbb{Z}/5\mathbb{Z} \times \mathbb{Z}^3 \times \mathbb{C}$$

Conductor

The conductor of a Hecke character is the smallest $\mathfrak{m}$ such that $\chi \in \mathrm{Hom}(C_{\mathfrak{m}}, \mathbb{C}^\times)$.

We represent a character χ by its column vector of coordinates corresponding to `gc.cyc`.

```
? chi = [0,0,0,5,0.1*I]~;
? gcharconductor(gc,chi)
% = [[5,4,1,4;0,1,0,0;0,0,1,0;0,0,0,1], []]
? gcharconductor(gc,4*chi)
% = [1,[]]
```

χ has conductor $\mathfrak{p}_5$ and χ^4 has trivial conductor.

Evaluation

Let $\mathfrak{p}$ be a prime of K and $\pi_{\mathfrak{p}}$ a uniformiser of $K_{\mathfrak{p}}$. Using the map $K_{\mathfrak{p}}^{\times} \to \mathbb{A}_K^{\times}$, we can evaluate χ on $K_{\mathfrak{p}}^{\times}$. Define

$$\chi(\mathfrak{p}) = \chi(\pi_{\mathfrak{p}}).$$

This is well-defined up to $\chi(\mathbb{Z}_{\mathfrak{p}}^{\times})$, which is a finite group. If $\mathfrak{p}$ does not divide the conductor of χ, it is well defined.

We evaluate Hecke characters with `gchareval`:

```
? pr11 = idealprimedec(bnf,11)[1];
? gchareval(gc,chi,pr11)
% = 0.8531383657 - 0.52168470249*I
```

Local characters: archimedean places

Let v be a place of K. We can restrict χ to $K_v^\times$.

Characters of $\mathbb{R}^\times$ are of the form

$$x \mapsto \operatorname{sign}(x)^k |x|^{i\varphi}$$

with $k \in \mathbb{Z}/2\mathbb{Z}$ and $\varphi \in \mathbb{C}$.

Characters of $\mathbb{C}^\times$ are of the form

$$z \mapsto \left(\frac{z}{|z|}\right)^k |z|^{2i\varphi}$$

with $k \in \mathbb{Z}$ and $\varphi \in \mathbb{C}$.

Local characters: archimedean places

We obtain the local characters with `gcharlocal`.

Archimedean places are represented by a number between 1 and $r_1 + r_2$.

```
? gcharlocal(gc,chi,1)
% = [5, -0.7160628256]
? gcharlocal(gc,chi,2)
% = [0, 0.9160628256]
```

Local characters: nonarchimedean places

Let $\mathfrak{p}$ be a prime of K.
A character on $K_{\mathfrak{p}}^{\times}$ is completely determined by

- its restriction to the finite group $\mathbb{Z}_{\mathfrak{p}}^{\times}/(\mathbb{Z}_{\mathfrak{p}}^{\times} \cap U(\mathfrak{m}))$, and
- its value $\exp(2\pi i\theta)$ on $\pi_{\mathfrak{p}}$.

Local characters: nonarchimedean places

We specify a nonarchimedean place by a prime ideal.

```
? pr5 = idealprimedec(bnf,5)[1];
? loc = gcharlocal(gc,chi,pr5,&bid)
% = [15, 0, 0, -0.15061499993]
? bid.cyc
% = [20, 5, 5]
? charorder(bid,loc[1..-2])
% = 4
```

We have $\mathbb{Z}_{\mathfrak{p}}^{\times}/(\mathbb{Z}_{\mathfrak{p}}^{\times} \cap U(\mathfrak{m})) \cong \mathbb{Z}/20\mathbb{Z} \times (\mathbb{Z}/5\mathbb{Z})^2$, and $\chi|_{\mathbb{Z}_{\mathfrak{p}}^{\times}}$ has order 4. So $\chi(\mathfrak{p})$ is well-defined up to multiplication by a 4-th root of unity.

L-function

Let χ be a Hecke character of conductor $\mathfrak{m}$. Define

$$L(\chi, s) = \prod_{\mathfrak{p} \nmid \mathfrak{m}} (1 - \chi(\mathfrak{p}) N(\mathfrak{p})^{-s})^{-1}.$$

This defines an L-function:

- it extends to a meromorphic function on $\mathbb{C}$;
- it satisfies a functional equation, with gamma factors given by the (k_v, φ_v) at archimedean places, and of conductor $|\Delta_K| N(\mathfrak{m})$.

L-function

We can use the `lfun` functionalities for L-functions of Hecke characters (currently: no imaginary component in χ).

```
? L = lfuncreate([gc,chi[1..-2]]);
? lfunparams(L)[1] \\conductor
% = 625
? lfunparams(L)[3]*1.
% = [5/2 - 0.8160628256*I, 0.8160628256*I,
    7/2 - 0.8160628256*I, 1 + 0.8160628256*I]
? lfuncheckfeq(L)
% = -132
? lfun(L,1)
% = 1.0185518145 + 0.1382746268*I
```

Algebraic characters

A Hecke character is called **algebraic** if for every complex embedding σ, there exists p_σ, q_σ such that for all $z \in (K_\sigma^\times)^\circ$,

$$\chi(z) = z^{-p_\sigma} (\bar{z})^{-q_\sigma}.$$

We then say that χ is of **type** $((p_\sigma, q_\sigma))_\sigma$.

Equivalently, there exists a number field E such that for all $\mathfrak{p}$,

$$\chi(\mathfrak{p}) \in E^\times.$$

Algebraic characters

We can test the algebraicity of a character and compute its type with `gcharisalgebraic`:

```
? gcharisalgebraic(gc,chi)
% = 0
? chi2 = [0,1,0,0,0]~
? gcharisalgebraic(gc,chi2,&typ)
% = 1
? typ
% = [[-1, 1], [0, 0]]
? gcharlocal(gc,chi2,1)
% = [2, 0]
? gcharlocal(gc,chi2,2)
% = [0, 0]
```

χ is not algebraic, but χ_2 is algebraic of type $((-1,1),(0,0))$.

Algebraic characters

The set of algebraic characters of modulus $\mathfrak{m}$ is a finitely generated group.
We can compute a basis of this group with `gcharalgebraic`:

```
? gcharalgebraic(gc)
% = [1 0    0  0]
    [0 1    0  0]
    [0 0    1  0]
    [0 0    0  0]
    [0 0 -1/2 -1]
```

Algebraic characters

Every finite order Hecke character is algebraic, and the type of an algebraic character determines it up to multiplication by a finite order character.
We can search for an algebraic character of a given type with `gcharalgebraic(gc,type)`:

```
? gcharalgebraic(gc,[[1,2],[3,4]])
% = []
? gcharalgebraic(gc,[[2,-2],[-1,1]])
% = [[0, -1, 2, 0, 0]~]
```

There is no character of type $((1,2),(3,4))$, but we found a character of type $((2,-2),(-1,1))$.

Identification

We can look for a character given some information about its values or its local characters with `gcharidentify`.

```
? pr31 = idealprimedec(bnf,31)[1];
? gcharidentify(gc,[pr11,pr31],[0.261946,-0.497068])
% = [3, -77916, 53772, 206992]~
```

This is probably meaningless because the number of digits of the output is of the same order as the precision we had on the values.

Identification

We need to reduce the working precision:

```
? localprec(6); chi3=gcharidentify(gc,[pr11,pr31],
  [0.261946,-0.497068])
% = [0, -3, 2, 8]~
? gchareval(gc,chi3,pr11,0)
% = 0.26194591587002798940182987097135921818
? gchareval(gc,chi3,pr31,0)
% = -0.49706763230668562700776309783089085752
```

Identification

To ensure reliable identification, even with low precision, you need to provide all archimedean places and the values at a set of primes that generates the ray class group $\mathrm{Cl}_{\mathfrak{m}}(K)$.

```
? chi4 = gcharidentify(gc,[1,2,pr11],[[-26,-0.1],
  [13,0.1],0.])
% = [1, -7, 13, 1]~
? gcharlocal(gc,chi4,1)
% = [-26, -0.1632125651]
? gcharlocal(gc,chi4,2)
% = [13, 0.1632125651]
? gchareval(gc,chi4,pr11)
% = 0.9007070934 - 0.4344269003*I
```

Example: CM abelian surface

By CM theory, the L-function of every CM abelian varietie is a product of L-functions of algebraic Hecke characters.
Let's compute an example: consider the genus 2 curve

$$C\colon y^2 + x^3y = -2x^4 - 2x^3 + 2x^2 + 3x - 2$$

and let A be its Jacobian.

```
? C = [-2*x^4 - 2*x^3 + 2*x^2 + 3*x - 2, x^3];
? L = lfungenus2(C);
? lfunparams(L)
% = [28561, 2, [0, 0, 1, 1]]
? factor(lfunparams(L)[1])
% = [13 4]
```

A has good reduction outside 13.

Example: CM abelian surface

```
E = bnfinit(y^4 - y^3 + 2*y^2 + 4*y + 3, 1);
poldegree(nfsubfieldscm(E)[1])
% = 4
```

The maximal CM subfield of E has degree 4, i.e. E is a CM field. It is known that A has CM by E.
We would like an associated Hecke character.

```
? pr13 = idealprimedec(E,13)[1];
? gc2 = gcharinit(E,pr13);
? gc2.cyc
% = [3, 0, 0, 0, 0.E-57]
? chiC = [1, -1, -1, 0, -1/2]~
```

Example: CM abelian surface

```
? gcharisalgebraic(gc2,chiC,&typ)
? typ
% = [[1, 0], [1, 0]]
```

This is the type we expect for an algebraic Hecke character corresponding to an abelian variety.

```
? L2 = lfuncreate([gc2,chiC]);
? lfunparams(L2)
% = [28561, 2, [0, 0, 1, 1]]
? exponent(lfunan(L,1000)-lfunan(L2,1000))
% = -120
```

The L-functions match!

Example: density

For varying conductor, the possible parameters at infinity of Hecke characters are dense.

```
? gc3 = gcharinit(x^3-3*x+1,2^20);
? chiapprox = gcharidentify(gc3,[1,2,3],[[0,Pi],
  [0,exp(1)],[0,-Pi-exp(1)]])
% = [0, 1338253, 2033118]~
? gcharlocal(gc3,chiapprox,1)
% = [0, 3.141592238]
? gcharlocal(gc3,chiapprox,2)
% = [0, 2.718283147]
```

For this χ, we have $\varphi_1 \approx \pi$ and $\varphi_2 \approx e$!

Example: partially algebraic characters

The algebraicity of a Hecke character is almost equivalent to the vanishing of all φ_σ parameters.

```
? gc4 = gcharinit(x^4-5,1);
? gc4.cyc
% = [0, 0, 0, 0.E-57]
? chipart = [1,0,0,0]~;
? gcharlocal(gc4,chipart,1)
% = [0, 0.7290851962]
? gcharlocal(gc4,chipart,2)
% = [0, -0.7290851962]
? gcharlocal(gc4,chipart,3)
% = [-2, 0.E-95]
```

For this χ, we have $\varphi_1, \varphi_2 = 0$ but $\varphi_3 = 0$!

Questions ?

Have fun with GP !

Implementation based on
`https://inria.hal.science/hal-03795267`.